

IRT Models in SAS NL MIXED

Example data: 635 older adults (age 80-100) self-reporting on 7 items assessing the Instrumental Activities of Daily Living (IADL) as follows:

1. Housework (cleaning and laundry)
2. Bedmaking
3. Cooking
4. Everyday shopping
5. Getting to places outside of walking distance
6. Handling banking and other business
7. Using the telephone

Two versions of a response format were available:

Binary → 0 = “needs help”, 1 = “does not need help”
 Categorical → 0 = “can’t do it”, 1=“big problems”, 2=“some problems”, 3=“no problems”

Higher scores indicate greater function. We will look at each response format in turn.

These models were estimated using Mplus (see Examples 6 and 7a), and the results are similar.

The first step is to convert the data from the format in which each person’s responses are on a single row (as separate variables) to a “stacked” format in which each item for each person is a separate row. We then create dummy code indicators for each item that take the value of 1 if the item is on that line, or 0 otherwise.

SAS Code to read in data and run NL MIXED:

```
* Generic session options - set page and line size, no page breaks;
OPTIONS nonumber nodate nocenter pagesize=MAX linesize=120 formdlim='-';
* Kill default titles on output;
TITLE;

* Location for files to be used or saved to - CHANGE THIS;
%GLOBAL filesave; %LET filesave=F:\14_Psyc948\OCTO;
LIBNAME octo "&filesave.";

* Importing IADL binary item data and stacking into a single variable;
DATA diadl; SET octo.iadl; ARRAY adia(7) dial-dia7;
  DO i=1 TO 7; item=i; dia=adia(i); OUTPUT; END;
RUN;
* Importing categorical IADL item data and stacking into a single variable;
DATA ciadl; SET octo.iadl; ARRAY acia(7) cial-cia7;
  DO i=1 TO 7; item=i; cia=acia(i); OUTPUT; END;
RUN;
* Create dummy indicator variables for each binary item;
DATA diadl; SET diadl; ARRAY dummy (7) i1-i7;
  DO d=1 TO 7; IF item=d THEN dummy(d)=1; ELSE dummy(d)=0; END;
  DROP i d dial-dia7 cial-cia7;
RUN;
* Create dummy indicator variables for each categorical item;
DATA ciadl; SET ciadl; ARRAY dummy (7) i1-i7;
  DO d=1 TO 7; IF item=d THEN dummy(d)=1; ELSE dummy(d)=0; END;
  DROP i d dial-dia7 cial-cia7;
RUN;
```

The SAS data for one person are shown below, in which each item response is in a separate row in the “cia” column. The i1–i7 variables are dummy codes that we will use to give each item its own parameters.

	case	item	cia	i1	i2	i3	i4	i5	i6	i7
8	2	1	0	1	0	0	0	0	0	0
9	2	2	0	0	1	0	0	0	0	0
10	2	3	0	0	0	1	0	0	0	0
11	2	4	0	0	0	0	1	0	0	0
12	2	5	0	0	0	0	0	1	0	0
13	2	6	0	0	0	0	0	0	1	0
14	2	7	3	0	0	0	0	0	0	1

```

*****;
/* PROC NLMIXED syntax commands used below:

DATA gives file used for analysis -- udatafile is named above
ITDETAILS gives iteration history for examining convergence
METHOD=GAUSS specifies Gauss-Hermite quadrature (default)
TECHNIQUE=QUANEW specifies Quasi-Newton optimization (default)
TECHNIQUE=NEWRAP specifies Newton-Raphson optimization (also used)
GCONV=1e-12 tightens the criterion for model convergence
NOAD specifies non-adaptive quadrature for theta (use if have problems estimating)
QPOINTS=15 specifies 15 quadrature points for theta
    The more points, the longer but more precise the estimation
NOITPRINT suppresses iteration history print

PARMS must list ALL parameters to be estimated with start values
MODEL defines equation, where [outcome] is distributed [parameter]
RANDOM lists random effects with distribution, here normal [mean, variance]
    SUBJECT defines level-2 nesting (here, persons)
    OUT lists dataset random effects (thetas) saved to
ODS OUTPUT ParameterEstimates saves all estimates to named dataset
*/
*****;

TITLE "1PL Binary Rasch Model with Theta Variance=1";
PROC NLMIXED DATA=diadl METHOD=GAUSS TECHNIQUE=QUANEW GCONV=1e-12 QPOINTS=15; *NOAD;
* All model parameters must be listed here WITH start values;
* Add better start values if you want to reduce estimation time or difficulty;
    PARS b01-b07=0 a=1;
* A placeholder for item difficulty is created by multiplying each item difficulty
  by a dummy code for that item, such that we get 7 separate estimates;
    b = b01*i1 + b02*i2 + b03*i3 + b04*i4 + b05*i5 + b06*i6 + b07*i7;
* The model equation is defined here -- same a for all items;
    p = exp(a*(theta-b)) / (1+exp(a*(theta-b)));
* The response variable is defined here;
    MODEL dia ~ binary(p);
* The random intercept (theta) is defined here, saved to named dataset;
    RANDOM theta ~ normal(0,1) SUBJECT = case OUT=octo.Theta_1PLbinary;
* All parameter estimates saved to named dataset;
    ODS OUTPUT ParameterEstimates=octo.Item_1PLBinary;

```

RUN;

30 seconds later...

Binary 1-PL (Rasch) Model Syntax and Truncated Output:

1PL Binary Rasch Model with Theta Variance=1
The NLMIXED Procedure

	Specifications	
Data Set		WORK.DIADL
Dependent Variable		dia
Distribution for Dependent Variable		Binary
Random Effects		theta
Distribution for Random Effects		Normal
Subject Variable		case
Optimization Technique		Dual Quasi-Newton
Integration Method		Adaptive Gaussian Quadrature
Dimensions		
Observations Used	4367	
Observations Not Used	106	
Total Observations	4473	
Subjects	635	
Max Obs Per Subject	7	
Parameters	8	
Quadrature Points	15	

NOTE: GCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	2927.9
AIC (smaller is better)	2943.9
AICC (smaller is better)	2944.0
BIC (smaller is better)	2979.6

This -2LL value is very close to what Mplus provides in LL form (SAS multiples LL by -2 for you).

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
b01	-0.3761	0.05463	634	-6.89	<.0001	0.05	-0.4834	-0.2689	-1.95E-6
b02	-1.0562	0.06712	634	-15.74	<.0001	0.05	-1.1880	-0.9244	1.672E-6
b03	-0.7979	0.06045	634	-13.20	<.0001	0.05	-0.9166	-0.6792	5.264E-6
b04	-0.4369	0.05502	634	-7.94	<.0001	0.05	-0.5450	-0.3289	-6.5E-6
b05	-0.4308	0.05485	634	-7.85	<.0001	0.05	-0.5385	-0.3231	1.083E-8
b06	-0.6802	0.05775	634	-11.78	<.0001	0.05	-0.7937	-0.5668	1.892E-6
b07	-1.6939	0.09239	634	-18.33	<.0001	0.05	-1.8753	-1.5124	1.334E-6
a	4.4443	0.2783	634	15.97	<.0001	0.05	3.8978	4.9908	-3.72E-7

The common discrimination (slope) parameter is "a", which is 4.4443. This matches closely with the common discrimination parameter of 4.383 estimated by Mplus for the same model.

The item difficulties (location) parameters are the b's, which gives the latent trait location at which a response of 1 (instead of 0) has a probability of 50%. These also match very closely with the b's that were obtained from Mplus (b's converted from the thresholds, b=threshold/loading).

Mplus version:

<pre> TITLE: 1PL Binary Model under ML DATA: FILE IS ADL.dat; VARIABLE: NAMES ARE case dial-dia7 cial-cia7; USEVARIABLES ARE dial-dia7; CATEGORICAL ARE dial-dia7; MISSING ARE .; ANALYSIS: ESTIMATOR IS ML; LINK IS LOGIT; MODEL: ! Factor loadings all constrained equal in 1PL IADL dial-dia7* (1); ! Item thresholds all free [dial\$1-dia7\$1*]; ! Factor mean=0 and variance=1 for identification [IADL@0]; IADL@1; TESTS OF MODEL FIT Loglikelihood H0 Value -1464.457 Information Criteria Number of Free Parameters 8 Akaike (AIC) 2944.915 Bayesian (BIC) 2980.544 Sample-Size Adjusted BIC 2955.144 (n* = (n + 2) / 24) </pre>	<pre> IRT PARAMETERIZATION IN TWO-PARAMETER LOGISTIC METRIC WHERE THE LOGIT IS 1.7*DISCRIMINATION*(THETA - DIFFICULTY) Item Discriminations IADL BY DIA1 4.383 0.272 16.094 0.000 DIA2 4.383 0.272 16.094 0.000 DIA3 4.383 0.272 16.094 0.000 DIA4 4.383 0.272 16.094 0.000 DIA5 4.383 0.272 16.094 0.000 DIA6 4.383 0.272 16.094 0.000 DIA7 4.383 0.272 16.094 0.000 Item Difficulties DIA1\$1 -0.383 0.053 -7.182 0.000 DIA2\$1 -1.069 0.065 -16.331 0.000 DIA3\$1 -0.809 0.059 -13.761 0.000 DIA4\$1 -0.444 0.054 -8.302 0.000 DIA5\$1 -0.438 0.053 -8.209 0.000 DIA6\$1 -0.690 0.056 -12.303 0.000 DIA7\$1 -1.711 0.091 -18.758 0.000 </pre>
--	---

Binary 2-PL (Unequal Slopes) Model Syntax and Truncated Output:

```

TITLE "2PL Binary Model with Theta Variance=1";
PROC NL MIXED DATA=diadl METHOD=GAUSS TECHNIQUE=QUANEW GCONV=1e-12 QPOINTS=15; *NOAD;
* All model parameters must be listed here WITH start values;
* This time, this includes the discrimination parameters per item;
* Add better start values if you want to reduce estimation time or difficulty;
  PARMS b01-b07=0 a01-a07=1;
* A placeholder for item difficulty is created by multiplying each item difficulty
  by a dummy code for that item, such that we get 7 separate estimates;
* We do the same to get an item discrimination separately per item;
  b = b01*i1 + b02*i2 + b03*i3 + b04*i4 + b05*i5 + b06*i6 + b07*i7;
  a = a01*i1 + a02*i2 + a03*i3 + a04*i4 + a05*i5 + a06*i6 + a07*i7;
* The model equation is defined here -- now different a across items;
  p = exp(a*(theta-b)) / (1+exp(a*(theta-b)));
* The response variable is defined here;
  MODEL dia ~ binary(p);
* The random intercept (theta) is defined here, saved to named dataset;
  RANDOM theta ~ normal(0,1) SUBJECT = case OUT=octo.Theta_2PLbinary;
* All parameter estimates saved to named dataset;
  ODS OUTPUT ParameterEstimates=octo.Item_2PLBinary;

```

RUN;

2.5 minutes later...

2PL Binary Model with Theta Variance=1
The NLMIXED Procedure

```

Specifications
Data Set                WORK.DIADL
Dependent Variable      dia
Distribution for Dependent Variable  Binary
Random Effects          theta
Distribution for Random Effects      Normal
Subject Variable        case
Optimization Technique  Dual Quasi-Newton
Integration Method      Adaptive Gaussian
                        Quadrature

Dimensions
Observations Used      4367
Observations Not Used  106
Total Observations     4473
Subjects               635
Max Obs Per Subject    7
Parameters             14
Quadrature Points      15
    
```

NOTE: GCONV convergence criterion satisfied.

```

Fit Statistics
-2 Log Likelihood      2907.4
AIC (smaller is better) 2935.4
AICC (smaller is better) 2935.5
BIC (smaller is better) 2997.7
    
```

Fit Statistics from 1 PL Binary Model

```

-2 Log Likelihood      2927.9
AIC (smaller is better) 2943.9
AICC (smaller is better) 2944.0
BIC (smaller is better) 2979.6
    
```

2927.9 – 2907.4 = 20.5 on df=6 is $p = .0023$,
so the 2PL is an improvement over the 1PL
(we need different slopes across items)

				Parameter Estimates					
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
b01	-0.3733	0.05513	634	-6.77	<.0001	0.05	-0.4815	-0.2650	-2.56E-7
b02	-1.0308	0.06815	634	-15.12	<.0001	0.05	-1.1646	-0.8969	-9.62E-6
b03	-0.7899	0.06232	634	-12.68	<.0001	0.05	-0.9123	-0.6675	-2.41E-6
b04	-0.4067	0.05226	634	-7.78	<.0001	0.05	-0.5093	-0.3041	0.000022
b05	-0.4261	0.05557	634	-7.67	<.0001	0.05	-0.5353	-0.3170	-6.26E-6
b06	-0.6978	0.06267	634	-11.13	<.0001	0.05	-0.8209	-0.5747	-1.54E-6
b07	-1.7958	0.1259	634	-14.26	<.0001	0.05	-2.0431	-1.5486	-3.22E-6
a01	4.4232	0.5761	634	7.68	<.0001	0.05	3.2918	5.5546	-8.4E-8
a02	5.0241	0.8155	634	6.16	<.0001	0.05	3.4227	6.6256	6.695E-7
a03	4.3836	0.5779	634	7.59	<.0001	0.05	3.2488	5.5185	6.268E-7
a04	8.0493	1.9078	634	4.22	<.0001	0.05	4.3030	11.7956	-1.47E-7
a05	4.3946	0.5547	634	7.92	<.0001	0.05	3.3055	5.4838	1.798E-7
a06	3.5189	0.4120	634	8.54	<.0001	0.05	2.7099	4.3278	6.651E-7
a07	3.3206	0.6078	634	5.46	<.0001	0.05	2.1271	4.5141	8.93E-7

The item discriminations (slope) parameters are the a's, which gives one slope per item. These match closely numerically to the discriminations estimated by Mplus.

The item difficulties (location) parameters are the b's, which gives the latent trait location at which a response of 1 (instead of 0) has a probability of 50%. These match fairly closely numerically to the b's estimated by Mplus.

Mplus version:

<pre> TITLE: 2PL Binary Model under ML DATA: FILE IS ADL.dat; VARIABLE: NAMES ARE case dial-dia7 cial-cia7; USEVARIABLES ARE dial-dia7; CATEGORICAL ARE dial-dia7; MISSING ARE .; ANALYSIS: ESTIMATOR IS ML; LINK IS LOGIT; MODEL: ! Factor loadings all free in 2PL IADL dial-dia7*; ! Item thresholds all free [dial\$1-dia7\$1*]; ! Factor mean=0 and variance=1 for identification [IADL@0]; IADL@1; TESTS OF MODEL FIT Loglikelihood H0 Value -1454.634 Information Criteria Number of Free Parameters 14 Akaike (AIC) 2937.268 Bayesian (BIC) 2999.619 Sample-Size Adjusted BIC 2955.170 (n* = (n + 2) / 24) </pre>	<pre> IRT PARAMETERIZATION IN TWO-PARAMETER LOGISTIC METRIC WHERE THE LOGIT IS 1.7*DISCRIMINATION*(THETA - DIFFICULTY) Item Discriminations IADL BY DIA1 4.328 0.560 7.725 0.000 DIA2 4.978 0.808 6.159 0.000 DIA3 4.323 0.570 7.579 0.000 DIA4 7.511 1.696 4.429 0.000 DIA5 4.248 0.527 8.062 0.000 DIA6 3.451 0.401 8.600 0.000 DIA7 3.283 0.601 5.467 0.000 Item Difficulties DIA1\$1 -0.376 0.052 -7.298 0.000 DIA2\$1 -1.045 0.065 -15.978 0.000 DIA3\$1 -0.801 0.059 -13.562 0.000 DIA4\$1 -0.415 0.047 -8.849 0.000 DIA5\$1 -0.432 0.052 -8.296 0.000 DIA6\$1 -0.708 0.060 -11.889 0.000 DIA7\$1 -1.816 0.126 -14.454 0.000 </pre>
---	---

Polytomous (4-category) Constrained Graded Response Model (Equal Slope across Items) Model Syntax and Truncated Output:

```
TITLE "1PL (constrained) Graded Response Model with Theta variance=1";
PROC NLMIXED DATA=ciadl METHOD=GAUSS TECHNIQUE=QUANEW GCONV=1e-12 QPOINTS=15; *NOAD;
* All model parameters must be listed here WITH start values;
* 3 difficulties per item, same discrimination across items;
* Add better start values if you want to reduce estimation time or difficulty;
  PARS b101-b107=-3 b201-b207=-2 b301-b307=-1 a=2;
  b1 = b101*i1 + b102*i2 + b103*i3 + b104*i4 + b105*i5 + b106*i6 + b107*i7;
  b2 = b201*i1 + b202*i2 + b203*i3 + b204*i4 + b205*i5 + b206*i6 + b207*i7;
  b3 = b301*i1 + b302*i2 + b303*i3 + b304*i4 + b305*i5 + b306*i6 + b307*i7;
* The cumulative model is now a series of equations conditional on response;
* eta1 refers to 0 vs. 123;
* eta2 refers to 01 vs. 23;
* eta3 refers to 012 vs. 3;
  eta1 = exp(a*(theta-b1)) / (1+exp(a*(theta-b1)));
  eta2 = exp(a*(theta-b2)) / (1+exp(a*(theta-b2)));
  eta3 = exp(a*(theta-b3)) / (1+exp(a*(theta-b3)));
* Calculate probabilities per category via subtraction of cumulative model;
  IF cia=0 THEN p = 1-eta1;
  ELSE IF cia=1 THEN p = eta1-eta2;
  ELSE IF cia=2 THEN p = eta2-eta3;
  ELSE IF cia=3 THEN p = eta3;
* The response variable is defined here;
* Because NLMIXED doesn't have a standard distribution for polytomous data,
we specify a general one as log(p);
  ll = log(p);
  MODEL cia ~ general(ll);
* The random intercept (theta) is defined here, saved to named dataset;
  RANDOM theta ~ normal(0,1) SUBJECT = case OUT=octo.Theta_1PLcategorical;
* All parameter estimates saved to named dataset;
  ODS OUTPUT ParameterEstimates=octo.Item_1PLcategorical;
RUN;
```

3 minutes later...

1PL (constrained) Graded Response Model with Theta variance=1
The NLMIXED Procedure

	Specifications	
Data Set		WORK.CIADL
Dependent Variable		cia
Distribution for Dependent Variable		General
Random Effects		theta
Distribution for Random Effects		Normal
Subject Variable		case
Optimization Technique		Dual Quasi-Newton
Integration Method		Adaptive Gaussian Quadrature

Dimensions

Observations Used	4230
Observations Not Used	243
Total Observations	4473
Subjects	634
Max Obs Per Subject	7
Parameters	22
Quadrature Points	15

NOTE: GCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	5182.4
AIC (smaller is better)	5226.4
AICC (smaller is better)	5226.6
BIC (smaller is better)	5324.3

Parameter	Estimate	Standard Error	DF	Parameter Estimates					
				t Value	Pr > t	Alpha	Lower	Upper	Gradient
b101	-1.5218	0.08581	633	-17.73	<.0001	0.05	-1.6903	-1.3533	6.944E-6
b102	-1.6077	0.08985	633	-17.89	<.0001	0.05	-1.7841	-1.4312	0.000047
b103	-1.5041	0.08610	633	-17.47	<.0001	0.05	-1.6732	-1.3350	-0.00005
b104	-1.3889	0.08101	633	-17.15	<.0001	0.05	-1.5480	-1.2299	9.125E-6
b105	-1.7737	0.09650	633	-18.38	<.0001	0.05	-1.9632	-1.5842	0.000015
b106	-1.7379	0.09717	633	-17.89	<.0001	0.05	-1.9287	-1.5470	-0.00002
b107	-2.3355	0.1508	633	-15.48	<.0001	0.05	-2.6317	-2.0393	-4.37E-6
b201	-1.0271	0.06694	633	-15.34	<.0001	0.05	-1.1585	-0.8956	-0.00002
b202	-1.2847	0.07571	633	-16.97	<.0001	0.05	-1.4334	-1.1361	-0.00001
b203	-1.1656	0.07207	633	-16.17	<.0001	0.05	-1.3072	-1.0241	0.000023
b204	-0.8760	0.06325	633	-13.85	<.0001	0.05	-1.0002	-0.7517	-0.00002
b205	-0.8440	0.06146	633	-13.73	<.0001	0.05	-0.9647	-0.7234	-0.00003
b206	-1.1431	0.07088	633	-16.13	<.0001	0.05	-1.2823	-1.0040	0.000021
b207	-1.8182	0.1012	633	-17.97	<.0001	0.05	-2.0170	-1.6195	-2.33E-6
b301	-0.1735	0.05471	633	-3.17	0.0016	0.05	-0.2809	-0.06604	0.000028
b302	-0.7613	0.05997	633	-12.69	<.0001	0.05	-0.8791	-0.6435	-0.00004
b303	-0.5765	0.05746	633	-10.03	<.0001	0.05	-0.6894	-0.4637	0.000019
b304	-0.2442	0.05444	633	-4.49	<.0001	0.05	-0.3511	-0.1373	0.00005
b305	-0.1806	0.05416	633	-3.33	0.0009	0.05	-0.2869	-0.07423	-0.00001
b306	-0.6691	0.05843	633	-11.45	<.0001	0.05	-0.7839	-0.5544	-0.00002
b307	-1.3076	0.07517	633	-17.40	<.0001	0.05	-1.4553	-1.1600	0.000015
a	3.9565	0.1935	633	20.45	<.0001	0.05	3.5765	4.3364	2.618E-6

The common item discrimination parameters is $a = 3.9565$. The item difficulties (location) parameters are the b 's, interpreted as the difficulty for each submodel. B1 is the item difficulty (where probability is 50%) of moving from a response of 0 to 123, B2 is 01 vs 23, and B3 is 012 vs 3.

Mplus does not provide IRT parameters from categorical response models, but they can be calculated from the item factor model loadings and thresholds it does provide. The b 's match pretty well, and a would have been 3.944 from Mplus.

Polytomous (4-category) Typical Graded Response Model (Unequal Slopes across Items) Model Syntax and Truncated Output:

```
TITLE "2PL (Typical) Graded Response Model with Theta variance=1";
PROC NL MIXED DATA=ciadl ITDETAILS METHOD=GAUSS TECHNIQUE=QUANEW GCONV=1e-12 QPOINTS=15; *NOAD;
* All model parameters must be listed here WITH start values;
* 3 difficulties per item, 1 discrimination per item now;
* Add better start values if you want to reduce estimation time or difficulty;
  PARS b101-b107=-3 b201-b207=-2 b301-b307=-1 a01-a07=2;
  b1 = b101*i1 + b102*i2 + b103*i3 + b104*i4 + b105*i5 + b106*i6 + b107*i7;
  b2 = b201*i1 + b202*i2 + b203*i3 + b204*i4 + b205*i5 + b206*i6 + b207*i7;
  b3 = b301*i1 + b302*i2 + b303*i3 + b304*i4 + b305*i5 + b306*i6 + b307*i7;
  a = a01*i1 + a02*i2 + a03*i3 + a04*i4 + a05*i5 + a06*i6 + a07*i7;
* The cumulative model is now a series of equations conditional on response;
* eta1 refers to 0 vs. 123;
* eta2 refers to 01 vs. 23;
* eta3 refers to 012 vs. 3;
  eta1 = exp(a*(theta-b1)) / (1+exp(a*(theta-b1)));
  eta2 = exp(a*(theta-b2)) / (1+exp(a*(theta-b2)));
  eta3 = exp(a*(theta-b3)) / (1+exp(a*(theta-b3)));
* Calculate probabilities per category via subtraction of cumulative model;
  IF cia=0 THEN p = 1-eta1;
  ELSE IF cia=1 THEN p = eta1-eta2;
  ELSE IF cia=2 THEN p = eta2-eta3;
  ELSE IF cia=3 THEN p = eta3;
* The response variable is defined here;
* Because NL MIXED doesn't have a standard distribution for polytomous data,
we specify a general one as log(p);
  ll = log(p);
  MODEL cia ~ general(ll);
* The random intercept (theta) is defined here, saved to named dataset;
  RANDOM theta ~ normal(0,1) SUBJECT = case OUT=octo.Theta_2PLcategorical;
* All parameter estimates saved to named dataset;
  ODS OUTPUT ParameterEstimates=octo.Item_2PLcategorical;
RUN;
```

12 minutes later...

2PL (Typical) Graded Response Model with Theta variance=1
The NLMIXED Procedure

```

Specifications
Data Set                WORK.CIADL
Dependent Variable      cia
Distribution for Dependent Variable  General
Random Effects          theta
Distribution for Random Effects      Normal
Subject Variable        case
Optimization Technique  Dual Quasi-Newton
Integration Method       Adaptive Gaussian
                        Quadrature
    
```

```

Dimensions
Observations Used      4230
Observations Not Used  243
Total Observations     4473
Subjects               634
Max Obs Per Subject    7
Parameters             28
Quadrature Points      15
    
```

NOTE: GCONV convergence criterion satisfied.

Fit Statistics

```

-2 Log Likelihood      5045.4
AIC (smaller is better) 5101.4
AICC (smaller is better) 5101.8
BIC (smaller is better) 5226.1
    
```

Fit Statistics for 1PL version of GRM

```

-2 Log Likelihood      5182.4
AIC (smaller is better) 5226.4
AICC (smaller is better) 5226.6
BIC (smaller is better) 5324.3
    
```

5182.4 – 5045.4 = 137.0 on df=6 is $p < .0001$, so the full GRM is an improvement over the constrained GRM (we need different slopes across items).

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
b101	-1.4127	0.08087	633	-17.47	<.0001	0.05	-1.5715	-1.2539	0.000056
b102	-1.5482	0.08953	633	-17.29	<.0001	0.05	-1.7240	-1.3724	0.000086
b103	-1.4644	0.08777	633	-16.68	<.0001	0.05	-1.6367	-1.2920	9.673E-6
b104	-1.2883	0.07796	633	-16.52	<.0001	0.05	-1.4414	-1.1352	8.937E-6
b105	-1.8321	0.1057	633	-17.33	<.0001	0.05	-2.0398	-1.6245	0.000019
b106	-1.8936	0.1209	633	-15.66	<.0001	0.05	-2.1310	-1.6563	-0.00011
b107	-3.2528	0.3204	633	-10.15	<.0001	0.05	-3.8820	-2.6235	-0.00003
b201	-0.9244	0.06116	633	-15.11	<.0001	0.05	-1.0445	-0.8043	-4.8E-6
b202	-1.1958	0.07377	633	-16.21	<.0001	0.05	-1.3407	-1.0510	-0.00002
b203	-1.1073	0.07191	633	-15.40	<.0001	0.05	-1.2485	-0.9661	-0.00008
b204	-0.7943	0.05969	633	-13.31	<.0001	0.05	-0.9115	-0.6771	8.442E-6
b205	-0.8375	0.06492	633	-12.90	<.0001	0.05	-0.9650	-0.7100	-0.00004
b206	-1.2189	0.08445	633	-14.43	<.0001	0.05	-1.3847	-1.0531	0.00005
b207	-2.4576	0.2159	633	-11.38	<.0001	0.05	-2.8815	-2.0336	0.000019
b301	-0.1611	0.05071	633	-3.18	0.0016	0.05	-0.2607	-0.06152	-0.0002
b302	-0.7006	0.05693	633	-12.31	<.0001	0.05	-0.8124	-0.5888	-0.00005
b303	-0.5394	0.05563	633	-9.70	<.0001	0.05	-0.6486	-0.4301	-0.00005
b304	-0.2314	0.05153	633	-4.49	<.0001	0.05	-0.3326	-0.1302	0.00012
b305	-0.1746	0.05478	633	-3.19	0.0015	0.05	-0.2821	-0.06701	0.00006
b306	-0.6878	0.06483	633	-10.61	<.0001	0.05	-0.8151	-0.5604	0.000072
b307	-1.6427	0.1367	633	-12.02	<.0001	0.05	-1.9111	-1.3743	0.000023
a01	6.9540	0.8559	633	8.13	<.0001	0.05	5.2734	8.6347	-3.03E-6
a02	5.2004	0.5529	633	9.41	<.0001	0.05	4.1146	6.2862	-0.00002
a03	4.6182	0.4538	633	10.18	<.0001	0.05	3.7271	5.5094	-0.00002
a04	5.7331	0.6122	633	9.36	<.0001	0.05	4.5309	6.9353	0.000016
a05	3.5579	0.2982	633	11.93	<.0001	0.05	2.9723	4.1435	5.281E-6
a06	2.8974	0.2605	633	11.12	<.0001	0.05	2.3858	3.4090	0.000016
a07	1.7762	0.2085	633	8.52	<.0001	0.05	1.3668	2.1856	0.000018

Each item now has a separate "a" discrimination parameter. The item difficulties (location) parameters are the b's, interpreted as the difficulty for each submodel. B1 is the item difficulty (where probability is 50%) of moving from a response of 0 to 123, B2 is 01 vs 23, and B3 is 012 vs 3.

Mplus does not provide IRT parameters from categorical response models, but they can be calculated from the item factor model loadings and thresholds it does provide. The a's and b's match pretty well the converted estimates from Mplus.

Polytomous (4-category) Partial Credit Model (Equal Slopes) Model Syntax and Truncated Output:

```

TITLE "1PL (typical) Partial Credit Model Model with Theta variance=1";
PROC NL MIXED DATA=ciadl METHOD=GAUSS TECHNIQUE=QUANEW GCONV=1e-12 QPOINTS=15; *NOAD;
* All model parameters must be listed here WITH start values;
* 3 step thresholds per item, same discrimination across items;
  PARMS d101-d107=-2 d201-d207=-1 d301-d307=0 a=2;
  d1 = d101*i1 + d102*i2 + d103*i3 + d104*i4 + d105*i5 + d106*i6 + d107*i7;
  d2 = d201*i1 + d202*i2 + d203*i3 + d204*i4 + d205*i5 + d206*i6 + d207*i7;
  d3 = d301*i1 + d302*i2 + d303*i3 + d304*i4 + d305*i5 + d306*i6 + d307*i7;
* The adjacent-category model is also a series of equations conditional on response;
* eta1 refers to 0 vs. 1;
* eta2 refers to 1 vs. 2;
* eta3 refers to 2 vs. 3;
  eta1 = exp(a*((theta-d1)));
  eta2 = exp(a*((theta-d1)+(theta-d2)));
  eta3 = exp(a*((theta-d1)+(theta-d2)+(theta-d3)));
* Probabilities per category estimated directly;
  IF cia=0 THEN p = 1 / (1 + eta1 + eta2 + eta3);
  ELSE IF cia=1 THEN p = eta1 / (1 + eta1 + eta2 + eta3);
  ELSE IF cia=2 THEN p = eta2 / (1 + eta1 + eta2 + eta3);
  ELSE IF cia=3 THEN p = eta3 / (1 + eta1 + eta2 + eta3);
* The response variable is defined here;
* Because NL MIXED doesn't have a standard distribution for polytomous data,
we specify a general one as log(p);
  ll = log(p);
  MODEL cia ~ general(ll);
* The random intercept (theta) is defined here, saved to named dataset;
  RANDOM theta ~ normal(0,1) SUBJECT = case OUT=octo.Theta_1PCM;
* All parameter estimates saved to named dataset;
  ODS OUTPUT ParameterEstimates=octo.Item_1PCM;
RUN;

```

5 minutes later...

1PL (typical) Partial Credit Model Model with Theta variance=1
The NL MIXED Procedure

Specifications

Data Set	WORK.CIADL
Dependent Variable	cia
Distribution for Dependent Variable	General
Random Effects	theta
Distribution for Random Effects	Normal
Subject Variable	case
Optimization Technique	Dual Quasi-Newton
Integration Method	Adaptive Gaussian Quadrature

Dimensions

Observations Used	4230
Observations Not Used	243
Total Observations	4473
Subjects	634
Max Obs Per Subject	7
Parameters	22
Quadrature Points	15

Fit Statistics

-2 Log Likelihood	5161.6
AIC (smaller is better)	5205.6
AICC (smaller is better)	5205.8
BIC (smaller is better)	5303.5

NOTE: GCONV convergence criterion satisfied.

Parameter	Estimate	Standard Error	DF	Parameter Estimates					
				t Value	Pr > t	Alpha	Lower	Upper	Gradient
d101	-1.3104	0.09796	633	-13.38	<.0001	0.05	-1.5028	-1.1181	1.243E-6
d102	-1.2928	0.1107	633	-11.68	<.0001	0.05	-1.5102	-1.0754	0.000013
d103	-1.2077	0.1045	633	-11.56	<.0001	0.05	-1.4129	-1.0025	3.661E-6
d104	-1.2004	0.09198	633	-13.05	<.0001	0.05	-1.3810	-1.0198	-1.88E-6
d105	-1.7265	0.1051	633	-16.43	<.0001	0.05	-1.9329	-1.5201	-0.00003
d106	-1.6211	0.1098	633	-14.77	<.0001	0.05	-1.8367	-1.4055	3.345E-7
d107	-2.2538	0.1879	633	-11.99	<.0001	0.05	-2.6228	-1.8847	-2.89E-8
d201	-1.1278	0.08263	633	-13.65	<.0001	0.05	-1.2900	-0.9655	-0.00002
d202	-1.3612	0.1036	633	-13.14	<.0001	0.05	-1.5646	-1.1577	5.059E-6
d203	-1.2661	0.09706	633	-13.04	<.0001	0.05	-1.4567	-1.0755	-0.00001
d204	-0.9251	0.07890	633	-11.73	<.0001	0.05	-1.0801	-0.7702	8.728E-7
d205	-0.8135	0.07010	633	-11.60	<.0001	0.05	-0.9511	-0.6758	-2.27E-6
d206	-1.0849	0.08619	633	-12.59	<.0001	0.05	-1.2541	-0.9156	0.000016
d207	-1.8568	0.1285	633	-14.45	<.0001	0.05	-2.1091	-1.6045	3.929E-6
d301	-0.2453	0.06003	633	-4.09	<.0001	0.05	-0.3632	-0.1274	0.000019
d302	-0.9249	0.07138	633	-12.96	<.0001	0.05	-1.0651	-0.7848	-0.00002
d303	-0.7038	0.06659	633	-10.57	<.0001	0.05	-0.8346	-0.5731	0.000019
d304	-0.3653	0.06186	633	-5.91	<.0001	0.05	-0.4868	-0.2438	-0.00002
d305	-0.2778	0.06045	633	-4.60	<.0001	0.05	-0.3965	-0.1591	-8.09E-6
d306	-0.8312	0.06971	633	-11.92	<.0001	0.05	-0.9680	-0.6943	8.535E-6
d307	-1.4324	0.08674	633	-16.51	<.0001	0.05	-1.6028	-1.2621	9.511E-6
a	3.1099	0.1650	633	18.85	<.0001	0.05	2.7859	3.4340	-4.32E-6

The common discrimination (slope) parameter is "a", which is 3.1099. The item step parameters (locations) are the d's. The d1's give the latent trait location at which, given the answer was 0 or 1, 1 becomes more likely than 0 (not 50%). The d2's give the latent trait location at which, given that the answer was 1 or 2, 2 becomes more likely than 1 (not 50%). The d3's give the latent trait location at which, given that the answer was 2 or 3, 3 becomes more likely than 1 (not 50%). This model is not currently estimable in Mplus.

Polytomous (4-category) Generalized Partial Credit Model (Unequal Slopes) Model Syntax and Truncated Output:

```
TITLE "2PL (generalized) Partial Credit Model Model with Theta variance=1";
PROC NL MIXED DATA=ciadl METHOD=GAUSS TECHNIQUE=QUANEW GCONV=1e-12 QPOINTS=15; *NOAD;
* All model parameters must be listed here WITH start values;
* 2 step thresholds per item, 1 discrimination per item;
PARMS d101-d107=-2 d201-d207=-1 d301-d307=0 a01-a07=2;
d1 = d101*i1 + d102*i2 + d103*i3 + d104*i4 + d105*i5 + d106*i6 + d107*i7;
d2 = d201*i1 + d202*i2 + d203*i3 + d204*i4 + d205*i5 + d206*i6 + d207*i7;
d3 = d301*i1 + d302*i2 + d303*i3 + d304*i4 + d305*i5 + d306*i6 + d307*i7;
a = a01*i1 + a02*i2 + a03*i3 + a04*i4 + a05*i5 + a06*i6 + a07*i7;
* The adjacent-category model is also a series of equations conditional on response;
* eta1 refers to 0 vs. 1;
* eta2 refers to 1 vs. 2;
* eta3 refers to 2 vs. 3;
eta1 = exp(a*((theta-d1)));
eta2 = exp(a*((theta-d1)+(theta-d2)));
eta3 = exp(a*((theta-d1)+(theta-d2)+(theta-d3)));
* Probabilities per category estimated directly;
IF cia=0 THEN p = 1 / (1 + eta1 + eta2 + eta3);
ELSE IF cia=1 THEN p = eta1 / (1 + eta1 + eta2 + eta3);
ELSE IF cia=2 THEN p = eta2 / (1 + eta1 + eta2 + eta3);
ELSE IF cia=3 THEN p = eta3 / (1 + eta1 + eta2 + eta3);
* The response variable is defined here;
* Because NL MIXED doesn't have a standard distribution for polytomous data,
we specify a general one as log(p);
ll = log(p);
MODEL cia ~ general(ll);
* The random intercept (theta) is defined here, saved to named dataset;
RANDOM theta ~ normal(0,1) SUBJECT = case OUT=octo.Theta_2PCM;
* All parameter estimates saved to named dataset;
ODS OUTPUT ParameterEstimates=octo.Item_2PCM;
RUN;
```

13 minutes later...

2PL (generalized) Partial Credit Model Model with Theta variance=1
The NL MIXED Procedure

Specifications

```
Data Set                WORK.CIADL
Dependent Variable     cia
Distribution for Dependent Variable   General
Random Effects        theta
Distribution for Random Effects      Normal
Subject Variable      case
Optimization Technique  Dual Quasi-Newton
Integration Method     Adaptive Gaussian
                        Quadrature
```

Dimensions

```
Observations Used      4230
Observations Not Used  243
Total Observations    4473
Subjects              634
Max Obs Per Subject   7
Parameters            28
Quadrature Points     15
```

NOTE: GCONV convergence criterion satisfied.

Fit Statistics

```
-2 Log Likelihood      5040.2
AIC (smaller is better) 5096.2
AIACC (smaller is better) 5096.6
BIC (smaller is better) 5220.9
```

Fit Statistics for 1PL version of PCM	
-2 Log Likelihood	5161.6
AIC (smaller is better)	5205.6
AIACC (smaller is better)	5205.8
BIC (smaller is better)	5303.5

5161.6 – 5040.2 = 121.4 on df=6 is $p < .0001$, so the generalized PCM is an improvement over the original PCM (we need different slopes across items).

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
d101	-1.3541	0.08105	633	-16.71	<.0001	0.05	-1.5133	-1.1949	-0.00002
d102	-1.3788	0.1024	633	-13.47	<.0001	0.05	-1.5799	-1.1778	-7.31E-6
d103	-1.2468	0.1040	633	-11.98	<.0001	0.05	-1.4511	-1.0425	-3.04E-6
d104	-1.2214	0.07968	633	-15.33	<.0001	0.05	-1.3779	-1.0649	8.432E-6
d105	-1.7791	0.1132	633	-15.72	<.0001	0.05	-2.0013	-1.5568	2.434E-6
d106	-1.6761	0.1449	633	-11.57	<.0001	0.05	-1.9606	-1.3916	4.1E-6
d107	-2.4570	0.3529	633	-6.96	<.0001	0.05	-3.1500	-1.7640	1.349E-6
d201	-0.9589	0.06569	633	-14.60	<.0001	0.05	-1.0879	-0.8299	-4.4E-6
d202	-1.2677	0.09304	633	-13.62	<.0001	0.05	-1.4504	-1.0850	6.519E-6
d203	-1.1998	0.09550	633	-12.56	<.0001	0.05	-1.3873	-1.0123	-5.9E-7
d204	-0.8352	0.06601	633	-12.65	<.0001	0.05	-0.9648	-0.7056	2.538E-6
d205	-0.7710	0.07235	633	-10.66	<.0001	0.05	-0.9131	-0.6290	1.566E-6
d206	-1.0386	0.1129	633	-9.20	<.0001	0.05	-1.2603	-0.8169	8.739E-6
d207	-2.0103	0.2360	633	-8.52	<.0001	0.05	-2.4737	-1.5469	-1.12E-6
d301	-0.1806	0.05193	633	-3.48	0.0005	0.05	-0.2826	-0.07867	0.000024
d302	-0.8082	0.06927	633	-11.67	<.0001	0.05	-0.9442	-0.6722	-3.24E-6
d303	-0.6484	0.06906	633	-9.39	<.0001	0.05	-0.7840	-0.5128	-8.32E-6
d304	-0.2783	0.05580	633	-4.99	<.0001	0.05	-0.3878	-0.1687	-5.56E-6
d305	-0.2855	0.06541	633	-4.37	<.0001	0.05	-0.4140	-0.1571	-0.00001
d306	-1.0383	0.1148	633	-9.04	<.0001	0.05	-1.2638	-0.8128	6.932E-6
d307	-2.1917	0.2475	633	-8.85	<.0001	0.05	-2.6777	-1.7056	-3.82E-6
a01	6.2340	0.9148	633	6.81	<.0001	0.05	4.4377	8.0303	7.454E-7
a02	4.0657	0.5217	633	7.79	<.0001	0.05	3.0412	5.0903	4.038E-8
a03	3.4489	0.4100	633	8.41	<.0001	0.05	2.6437	4.2540	1.91E-6
a04	4.9725	0.6752	633	7.36	<.0001	0.05	3.6467	6.2983	1.188E-6
a05	2.8899	0.2926	633	9.88	<.0001	0.05	2.3154	3.4645	5.719E-7
a06	2.0018	0.2213	633	9.05	<.0001	0.05	1.5673	2.4363	-1.8E-6
a07	1.3368	0.1800	633	7.43	<.0001	0.05	0.9834	1.6903	9.924E-6

The item discriminations (slope) parameters are the a's, which gives one slope per item. The item step parameters (locations) are the d's, interpreted as before. This model is also not currently estimable in Mplus.



Finally, we can compare information criteria to decide which fits better, the graded response model or the generalized partial credit model. They both use 28 parameters (unequal slopes):

Fit Statistics from GRM

```
-2 Log Likelihood      5045.4
AIC (smaller is better) 5101.4
AIACC (smaller is better) 5101.8
BIC (smaller is better) 5226.1
```

The AIC and BIC are slightly smaller for the generalized PCM, so apparently that one is better.

Comparison of Prediction from Graded Response vs. Generalized Partial Credit Model for Item 3:



The GRM and GPCM make very similar predictions here (and often in general). Further, both models suggest there is little distinction between 0 (“can’t do it”) and 1 (“big problems doing it”) for item 3.